

SED1520 液晶控制器的应用

目 录

一、 <i>SED1520</i> 的引脚功能及控制时序	2
1、 <i>SED1520</i> 的总体特点	2
2、 <i>SED1520</i> 的部分引脚功能介绍	3
3、 <i>SED1520</i> 的读 / 写操作时序	3
4、 <i>SED1520</i> 电特性	3
5、 <i>SED1520</i> 显示 RAM 的结构	4
二、指令系统	4
1、读状态指令	4
2、复位	4
3、占空比选择	4
4、显示起始行设置	4
5、终止驱动开/关选择	4
6、ADC 选择指令	5
7、显示开 / 关指令	5
8、设置页地址	5
9、设置列地址	5
10、改写方式设置指令	5
11、改写结束指令	5
12、写数据	5
13、读数据	5
三、液晶模块 <i>MGLS-12032A/B</i> 和 <i>MGLS-10032A/B</i> 的电路结构特点	5
四、液晶模块 <i>MGLS10032</i> 和 <i>MGLS-12032A</i> 的应用	6
1、直接访问方式接口电路及驱动程序	6
2、间接控制方式接口电路及驱动程序	8
3、应用子程序	9





北京德彼克创新科技有限公司

地址：北京海淀区中关村南大街 48 号 邮编：100081 技术：010-82190029/31 /62175129-805, 806
业务：010-82190024/25/26/27 82190029/31 /62175129-801, 802 传真：82190028 , 62175129
E-Mail: dpksales@public.bta.net.cn 网址：www.D-Peak.com www.D-Peak.com.cn www.LCD.net.cn

SED1520 液晶显示驱动器的应用

SED1520 液晶显示驱动器是一种点阵图形式液晶显示驱动器，它可直接与 8 位微处理器相连，集行、列驱动器和控制器于一体，因此使用起来十分方便，作为内藏式控制器被广泛应用于小规模液晶显示模块(例如香港精电[Varitronix]公司的 MGLS-10032A/B、12032A/B，台湾 DATA VISION 公司的 DG-12816/12232 等)。本文将介绍 SED1520 的结构、功能，详细叙述以 SED1520 为控制器的液晶模块的使用。



Pin Name	Function
DB0 - DB7	Data input
A0	Selects display data or instructions. HIGH: Display data. LOW: Instructions.
RES	Resets the system and selects the interface type for a 68-port/80-port MPU. HIGH: 68-port MPU interface. LOW: 80-port MPU interface.
CS	Chip Select input. LOW: Active level sensing.
E (RD)	Read/Write Enable signal when a 68-port MPU is connected. (Active-LOW Read Enable signal when an 80-port MPU is connected.)
R/W (WR)	Read/Write Select signal when a 68-port MPU is connected. HIGH: Read Select LOW: Write Select (Active-LOW Write Enable input when an 80-port MPU is connected. Rising edge sensing.)
CL	External clock input (only effective with external clock types).
FR	LCD Frame (AC-conversion) signal input/output.
SEG0 - SEG60	Segment output for driving the LCD.
COM0 - COM15 (COM31 - COM16)	Common output for driving the LCD.
M/S	Master/Slave Select signal.
Vcc	5V power supply.
Vss	0V power supply (GND level).
V1, V2, V3, V4, V5	Power supplies for driving the LCD. V ₀₀ ≥ V1 ≥ V2 ≥ V3 ≥ V4 ≥ V5

一、SED1520 的引脚功能及控制时序

1、SED1520 的总体特点

(1) 内部 RAM 中的 1 位数据控制液晶屏上一个点的亮灭状态。“1”表示亮，“0”表示暗。SED1520 内部 RAM 容量为 2560（与屏幕的对应关系见第二节）。

(2) 具有 16 个行驱动口和 61 个列驱动口。

(3) 可直接与 80 系列微处理器相连，亦可直接与 68 系列微处理器相连（由引脚 RES 的状态决定）。

(4) 驱动占空比为 1/16 或 1/32。

(5) 可以与 SED1521 配合使用，以便扩展列驱动口数目。

(6) 数据总线最高速度为 10M。

2、SED1520 的部分引脚功能介绍

SED1520 有两种类型，即 SED1520FAA 和 SED1520FOA，其引脚定义基本相同，只是后者具有内部 18KHz 时钟。在 SED1520 的 100 个引脚中，大多数用于驱动液晶屏，与液晶模块的编程关系不大。现仅将与编程有关的引脚介绍如下：

(1) DB0—DB7，双向三态数据总线。

(2) A0：数据 / 指令选择。A0="1"选择数据通道；A0="0"选择指令通道。

(3) RES：必须注意的是，这个引脚要实现的功能是上电时低电平（负脉冲）使芯片复位。正常工作时决定 SED1520 与 MPU 的接口类型：“1”表示可直接与 68 系列微处理器相连；“0”表示可直接与 80 系列微处理器接口（复位由软件来控制，将在指令系统时介绍）。时序选择一般都由模块制造商确定，用户无法改变。

(4) CS (OSC1) 和 CL (OSC2)：对 SED1520FAA 来说，CS 作为片选端，而 CL 是外接时钟输入端；对 SED1520FOA，CL 是外接时钟信号的输入端，但 OSC1 和 OSC2 之间接一电阻，可以启动内部 18KHz 时钟。OSC2 还作为级联使用时时序的输入输出端。主方式下，OSC2 输出时钟信号；从方式下，OSC2 输入时钟信号。

(5) E(RD)：与 68 系列 MPU 相连时 (RES=1), E 为读/写使能信号；与 80 系列 MPU 相连时 (RES=0), RD 为读操作信号。

(6) R/W(WR)：与 68 系列 MPU 相连时 (RES=1), R/W 为读 / 写选择信号；与 80 系列 MPU 相连时 (RES=0), WR 为写操作信号。

(7) FR：LCD 帧同步信号输入端：（当 M/S=1 时，该脚为输出端；当 M/S=0 时，该脚为输入端）

(8) SEG0—SEG60：LCD 列驱动输出端。


(9) COM0—COM15：LCD 行驱动输出端

(10) V1, V2, V3, V4, V5：LCD 驱动电源，要求： $VDD \geq V1 \geq V2 \geq V3 \geq V4 \geq V5$


(11) VDD：逻辑电源+5V

(12) VSS：逻辑电源地

3、SED1520 的读 / 写操作时序



5、SED1520 显示 RAM 的结构



二、指令系统

SED1520 液晶显示驱动器共有 13 种显示指令，下面以与 68 系列 MPU 接口为例 (RES=1)，分别介绍一下这 13 种指令。

1、读状态指令

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
1	0	BUSY	ADC	OFF/ON	RESET	0	0	0	0

当 SED1520 处于“忙”状态时，除了读状态指令，其它指令均不起任何作用，因此在向 SED1520 发指令或写数据时，一般都要先读一下状态，判断是否“忙”。

BUSY:	1: 忙状态	0: 准备状态
ADC:	1: 正常输出 (右向)	0: 反向输出 (左向) (具体见 ADC 选择)
OFF/ON:	1: 显示关闭	0: 显示打开
RESET:	1: 复位状态	0: 正常状态

2、复位

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	0	1	1	1	1/0

该指令为复位指令，执行该指令后，使显示起始行置为第 0 行，列地址置为 0，页地址置为 3。

3、占空比选择

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	0	1	0	0	0/1

D0=0 占空比为 1/16，D0=1 为 1/32。

驱动 32 行液晶显示时，使 D0 为 1；驱动 16 行时使 D0=0。

4、显示起始行设置

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	1	1	1	0	显示起始行 (1—31)				

该指令设置了对应显示屏上首行的显示 RAM 中的行号。有规律地修改行号，可实现上、下滚屏功能。

5、终止驱动开/关选择

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	0	0	1	0	0/1

该指令用软件终止 LCD 驱动的输出。使系统在不显示状态下停止对 LCD 的驱动输出，从而降低系统

的功耗。终止驱动指令须在关显示状态下输入。

D0=1 为终止驱动，D0=0 为正常驱动。

6、ADC 选择指令

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	0	0	0	0	0/1

该指令用来设置列驱动输出口与液晶屏的列驱动线的连接方式。应根据厂方提供的模块实际接线设置。一般设为 0。

7、显示开 / 关指令

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	0	1	1	1	0/1

D0=1 为开显示；D0=0 为关显示。该指令不影响显示 RAM 内容。

8、设置页地址

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	0	1	1	1	0	页地址(0—3)	

9、设置列地址

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	0 列地址(0—79)							

10、改写方式设置指令

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	1	1	0	0	0	0	0

该指令发出后，使得每次写数据后列地址自动增 1，而读数据后列地址仍保持原值不变。这种称为“改写模式” (Read Modify Write) 的方式为逐个读取像点，改写后写回的工作提供了方便。

11、改写结束指令

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	0	1	1	1	0	1	1	1	0

该指令执行后，将结束改写方式，以后无论读或写数据后，列地址都自动增 1。

12、写数据

R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
0	1	显示数据							

13、读数据


R/W	A0	D7	D6	D5	D4	D3	D2	D1	D0
1	1	显示数据							

正常状态下，写数据或读数据后，列地址将自动增 1。

三、液晶模块 MGLS-12032A/B 和 MGLS-10032A/B 的电路结构特点

MGLS-12032A/B 和 MGLS-10032A/B 这四种液晶模块都是由两片 SED1520 来驱动的。它们的结构基本上相同。MGLS-12032 A 和 B 的差别在于像素点及视屏尺寸不同。MGLS-10032A 和 B 的差别也是如此。MGLS-12032 和 MGLS-10032 之间的差别是前者液晶屏的点阵为 100 列，每个 SED1520 中的 SEG0-SEG60 都只用了 SEG0—SEG49 这 50 个列驱动口；而后者液晶屏的点阵为 120 列，每个 SED1520 都用了 SEG0—SEG59 这 60 个列驱动口。其余部分，这四种液晶模块是完全一致的。逻辑电路图如下：


可以看出，两个 SED1520 均为 SED1520FOA，所以其时钟可由内部产生，SED1520 (1) 为主方式工作，OSC1 和 OSC2 跨接一个电阻，并由 OSC2 输出时序信号；SED1520 (2) 为从方式工作，时钟由 SED1520 (1) 提供，从 OSC2 输入。两个芯片的 RES 上电后经一定时间升为高电平，所以控制时序为 68 系列 MPU 的时序。两个芯片的选通由 E1、E2 信号决定。



对于这几种模块，ADC 选择应设置为 0，占空比设置成 1/32。

四、液晶模块 MGLS10032 和 MGLS-12032A 的应用

1、直接访问方式接口电路及驱动程序



驱动子程序

(1) 写指令子程序

左半屏[SED1520 (1) —E1]

占用寄存器: A, R2, DPTR 输入寄存器: R2-指令代码

```

PR0:  MOV    DPTR, #8002H           ;读指令地址
      MOVX   A, @DPTR             ;读 BF
      JB     ACC.7, PR0
      MOV    DPTR, #8000H         ;写指令地址
      MOV    A, R2
      MOVX   @DPTR, A
    
```

RET

(2) 写数据程序

左半屏[SED1520 (1) —E1]

占用寄存器: A, DPTR, R2 输入寄存器: R2—数据

```

PR1:  MOV    DPTR, #8002H           ;读指令地址
      MOVX   A, @DPTR
      JB     ACC.7, PR1             ;判忙
      MOV    DPTR, #8001H         ;写数据地址
      MOV    A, R2
      MOVX   @DPTR, A
      RET
    
```

(3) 写指令子程序

左半屏[SED1520 (2) —E2]

占用寄存器: A, R2, DPTR 输入寄存器: R2—指令代码

```

PR2:  MOV    DPTR, #4002H         ;读指令地址
      MOVX   A, @DPTR             ;读 BF
      JB     ACC.7, PR2
      MOV    DPTR, #4000H         ;写指令地址
      MOV    A, R2
      MOVX   @DPTR, A
      RET
    
```

(4) 写数据程序


左半屏[SED1520 (2) —E2]

占用寄存器: A, DPTR, R2 输入寄存器: R2—数据

```

PR3:  MOV    DPTR, #4002H         ;读指令地址
      MOVX   A, @DPTR
      JB     ACC.7, PR3           ;判忙
      MOV    DPTR, #4001H         ;写数据地址
      MOV    A, R2
      MOVX   @DPTR, A
      RET
    
```

2、间接控制方式接口电路及驱动程序



驱动子程序

(1) 写指令子程序

左半屏[SED1520 (1) —E1]

占用寄存器: A, R2 输入寄存器: R2-指令代码

```

PR0:  CLR    P3.0                ; A0=0
      SETB   P3.1                ; R/W=1
      SETB   P3.2                ; E1=1
      MOV    A,P1                ; 读忙标志
      CLR    P3.2                ; E1=0
      JB     ACC.7, PR0          ; 判忙
      CLR    P3.1                ; R/W=0
      MOV    A, R2               ; 取指令代码
      MOV    P1, A
      SETB   P3.2                ; E1=1
      CLR    P3.2                ; E1=0
      RET

```

(2) 写数据子程序

左半屏[SED1520 (1) —E1]

占用寄存器: A, R2 输入寄存器: R2—指令代码

```

PR1:  CLR    P3.0                ; A0=0
      SETB   P3.1                ; R/W=1
      SETB   P3.2                ; E1=1
      MOV    A, P1                ; 读忙标志
      CLR    P3.2                ; E1=0
      JB     ACC.7, PR0          ; 判忙
      SETB   P3.0                ; A0=1
      CLR    P3.1                ; R/W=0
      MOV    A, R2               ; 取指令代码
      MOV    P1, A
      SETB   P3.2                ; E1=1
      CLR    P3.2                ; E1=0
      RET

```

(3) 写指令子程序

左半屏[SED1520 (2) —E2]

占用寄存器: A, R2 输入寄存器: R2-指令代码

```

PR2:  CLR    P3.0                ; A0=0
      SETB   P3.1                ; R/W=1
      SETB   P3.2                ; E2=1
      MOV    A, P1                ; 读忙标志
      CLR    P3.3                ; E2=0
      JB     ACC.7, PR0          ; 判忙
      CLR    P3.1                ; R/W=0
      MOV    A, R2               ; 取指令代码
      MOV    P1, A
      SETB   P3.3                ; E2=1

```

```

CLR    P3.3                ; E2=0
RET

```

(4) 写数据子程序

左半屏[SED1520 (2) —E2]

占用寄存器: A, R2 输入寄存器: R2—指令代码

```

PR3:  CLR    P3.0                ; A0=0
      SETB   P3.1                ; R/W=1
      SETB   P3.2                ; E2=1
      MOV    A, P1                ; 读忙标志
      CLR    P3.3                ; E2=0
      JB     ACC.7, PR0           ; 判忙
      SETB   P3.0; A0=1
      CLR    P3.1; R/W=0
      MOV    A, R2                ; 取指令代码
      MOV    P1, A
      SETB   P3.3                ; E2=1
      CLR    P3.3                ; E2=0
      RET

```

3、应用子程序

(1) 清显示 RAM 子程序

占用寄存器: A, R2, R3, R4

```

PR4:  MOV    R4, #00H
PR41: MOV    A, R4
      ORL    A, #0B8H
      MOV    R2, A                ; 页地址设置
      LCALL PR0
      LCALL PR2
      MOV    R2, #00H            ; 列地址设置
      LCALL PR0
      CLALL PR2
      MOV    R3, #40H
PR42: MOV    R2, #00H
      LCALL PR1
      LCALL PR3
      DJNZ   R3, PR42
      INC    R4
      CJNE   R4, #04H, PR41
      RET

```

(2) 初始化子程序

```

PR5:  MOV    R2, #0E2H            ; 复位
      LCALL PR0
      LCALL PR2                ; 禁止终止驱动方式
      MOV    R2, #0A4H
      LCALL PR0
      LCALL PR2                ; 占空比设置 1/32

```

```

MOV    R2, #0A9H
LCALL  PR0
LCALL  PR2
MOV    R2, #0A0H           ; ADC 选择正顺序输出
LCALL  PR0
LCALL  PR2
MOV    R2, #0AFH          ; 开显示
LCALL  PR0
LCALL  PR2
MOV    R2, #0C0H          ; 显示起始行设置
LCALL  PR0
LCALL  PR2
LCALL  PR4                ; 调清 RAM 子程序
RET
    
```

(3) 字符显示

如左图所示，可造出“A”的字模为如下八个字节

DB: 00H 00H 00H 7EH 11H 11H 11H 7EH

SED1520 显示 RAM 是以一行八列形式写入的，所需的字库数据格式如上图：

以上的八个字节均是以每列最下面一位作为 MSB，最上面一位作为 LSB，这正是 SED1520 建立字模所要求的，其它字符字模的建立同是如此。

SED1520 中 RAM 有 32 行，分为 4 页每页 8 行，这样每页就可以写一行字符，总共可显示 4 行字符。RAM 一页为 80 字节，MGLS10032 使用 50 个字节（列），MGLS12032 使用 60 个字节。按 8*8 点阵字符划分显示屏区域，我们将 MGLS10032 屏分成 12.5*4 字符块，分布如下：

行\列	0	1	2	3	4	5	6	7	8	9	A	B
0												
1												
2												
3												

|----- SED1520 (1) -----|
|----- SED1520 (2) -----|

其中每行第 7 个字符在 SED1520 (1) 占据 2 个字节，在 SED1520 (2) 占 6 个字节，在编程时要注意两个芯片 RAM 区的转换。

我们将 MGLS-12032 屏分成 15*4 字符块，分布如下：

行/列	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E
0															
1															
2															
3															

|----- SED1520 (1) -----|
|----- SED1520 (2) -----|

其中每行第八个字符在 SED1520 (1) 占 4 个字节，在 SED1520 (2) 占有 4 个字节，在编程时要注意两个芯片 RAM 区的转换。

写入门 8*8 点阵字符子程序：

占用寄存器: A, B, R0, R1, R2, R3, R4, R5, R6, DPTR, 60H, 61H

输入寄存器: 60H, 61H

60H: 高4位: 行号; 低4位: 列号

61H: 字符代码

```

PR6:  MOV    A, 60H
      ANL    A, #0FH
      MOV    R4, A           ; 列号
      MOV    A, 60H
      SWAP   A
      ANL    A, #03H       ; 行号
      ORL    A, #0B8H
      MOV    R3, A         ; 页设置
      CJNE   R4, #DB1, PR61 ; MGLS10032: DB1=06H;  MGLS12032: DB1=07H
      LJMP   PR6A
PR61: MOV    R6, #00H      ; SED1520 (1)
      JC     PR62
      MOV    A, R4
      SUBB   A, #DB2       ; DB2=DB1+1
      MOV    R4, A
      MOV    R6, #01H     ; SED1520 (2)
PR62: MOV    A, R4
      MOV    B, #08H
      MUL   AB
      MOV    R4, A         ; 确定列地址
      MOV    R2, A
      CJNE   R6, #00H, PR63
      LCALL  PR0           ; SED1520 (1)
      LJMP   PR64
PR63: MOV    A, #DB3      ; MGLS10032: DB3=02H;  MGLS12032: DB3=04H
      ADD   A, R2
      MOV    R2, A         ; 修改列地址
      LCALL  PR2           ; SED1520 (2)
PR64: MOV    A, R3
      MOV    R2, A         ; 页地址设置
      CJNE   R6, #00H, PR65
      LCALL  PR0           ; SED1520 (1)
      LJMP   PR66
PR65: LCALL  PR2           ; SED1520 (2)
PR66: MOV    R5, 00H      ; 间址寄存器
      MOV    DPTR, #TAB1  ; 字符库首地址
      MOV    A, 61H       ; 取字符代码
      MOV    B, #08H
      MUL   AB
      ADD   A, DPL
      MOV    R0, A
      MOV    A, B

```

```

        ADDC    A, DPH
        MOV     R1, A
PR67:   MOV     A, R0
        MOV     DPL, A
        MOV     A, R1
        MOV     DPH, A
        MOV     A, R5
        MOVC   A, @A+DPTR           ; 取字模
        MOV     R2, A
        CJNE   R6, #00H, PR68
        LCALL  PR1                   ; SED1520 (1)
        LJMP   PR69
PR68:   LCALL  PR3                   ; SED1520 (2)
PR69:   INC     R5
        CJNE   R5, #08H, PR67
        RET

```

中间结合部字符位字符显示处理

```

PR6A:   MOV     R2, #DB4              ; MGLS10032: DB3=32H   ; MGLS12032: DB3=38H
        LCALL  PR0                   ; SED1520 (1)
        MOV     R2, #00H
        LCALL  PR2                   ; SED1520 (2)
        MOV     A, R3                 ; 页设置
        MOV     R2, A
        LCALL  PR0                   ; SED1520 (1)
        LCALL  PR2                   ; SED1520 (2)
        MOV     R5, #00H              ; 间址寄存器
        MOV     DPTR, #TAB1           ; 字符库首地址
        MOV     A, 61H                 ; 取字符代码
        MOV     B, #08H
        MUL    AB
        ADD    A, DPL
        MOV     R0, A
        MOV     A, B
        ADDC   A, DPH
        MOV     R1, A
PR6A:   MOV     A, R0
        MOV     DPL, A
        MOV     A, R1
        MOV     DPH, A
        MOV     A, R1
        MOV     DPH, A
        MOV     A, R5
        MOVC   A, @A+DPTR           ; 取字模
        MOV     R2, A
        CJNE   R5, #DB3, PR6B

```


```

PR6B: JNC    PR6C
      LCALL  PR1                ; SED1520 (1)
      LJMP  PR6D
PR6C: LCALL  PR3                ; SED1520 (2)
PR6D: INC    R5
      CJNE  R5, #08H, PR6A
      RET
TAB:  DB...                    ; 字符库
    
```

(4) 显示汉字


显示汉字的基本方法跟显示字符的基本方法相符，所不同的是，汉字至少是 16×16 点阵，显示一个 16×16 点阵需送进 32 个字节，则必须向两页中对应列送。这样，MGLS10032 和 MGLS12032 可以显示两行汉字。

我们将 MGLS-10032 屏分成 6×2 汉字块，分布如下：



其中每行第四汉字在 SED1520 (1) 占据 2 个字节，在 SED1520 (2) 占 14 个字节，在编程时要注意两个芯片 RAM 区的转换。

我们将 MGLS-12032 屏分成 7×2 汉字块，分布如下：



其中每行第四个汉字在 SED1520 (1) 占有 12 个字节，在 SED1520 (2) 占 14 个字节，在编程时要注意两个芯片 RAM 区的转换。

注意：从计算机内提取的汉字数组格式下正好与 SED1520 的格式差 90 度，本公司提供的汉字字模提取程序已经完成了转换工作。

写入 16×16 点阵字子程序

占用寄存器：A, B, R0, R1, R2, R3, R4, R5, R6, DPTR, 60H, 61H

输入寄存器：60H, 61H

60H: 高 4 位：行号；低 4 位：列号

61H: 汉字代码

```

PR7:  MOV    A, 60H
      ANL    A, #07H
      MOV    R4, A                ; 列号
      MOV    A, 60H
      SWAP  A
      ANL    A, 01H              ; 行号
      RL    A
      ORL    A, #0B8H
      MOV    R3, A                ; 页设置
      CJNE  R4, #03H, PR71       ; 结合部字符位
      LJMP  PR7G
PR71: MOV    R6, #00H            ; SED1520 (1)
    
```

```

JC      PR72
MOV     A, R4
SUBB   A, #04H
MOV     R4, A
MOV     R6, 01H           ; SED1520 (2)
PR72:  MOV     A, R4
        MOV     B, #10H
        MUL     AB
        MOV     R4, A           ; 确定列地址
        MOV     R2, A
        CJNE    R6, #00H, PR73
        LCALL   PR0           ; SED1520 (1)
        LJMP    PR74
PR73:  MOV     A, #DB1           ; MGLS10032: DB1=0EH   ; MGLS12032: DB1=04H
        ADD     A, R2
        MOV     R2, A           ; 列地址修正
        MOV     R4, A
        LCALL   PR2           ; SED1520 (2)
PR74:  MOV     A, R3
        MOV     R2, A           ; 页地址设置
        CJNE    R6, #00H, PR75
        LCALL   PR0           ; SED1520 (1)
        LJMP    PR76
PR75:  LCALL   PR2           ; SED1520 (2)
PR76:  MOV     R5, 00H           ; 间址寄存器
        MOV     DPTR, #TAB2     ; 汉字库首地址
        MOV     A, 61H           ; 取汉字代码
        MOV     B, #20H
        MUL     AB
        ADD     A, DPL
        MOV     R0, A
        MOV     A, B
        ADDC    A, DPH
        MOV     R1, A
PR77:  MOV     A, R0
        MOV     DPL, A
        MOV     A, R1
        MOV     DPH, A
        MOV     A, R5
        MOVC   A, @A+DPTR       ; 取字模
        MOV     R2, A
        CJNE    R6, #00H, PR78
        LCALL   PR1           ; SED1520 (1)
        LJMP    PR79
PR78:  LCALL   PR3           ; SED1520 (2)
PR79:  INC     R5

```

```

        CJNE  R5, #10H, PR77          ; 上半部数组 (16 字节)
        MOV   A, R4                    ; 列地址设置
        MOV   R2, A
        CJME  R6, #00H, PR7A
        LCALL PR0                      ; SED1520 (1)
        LJMP  PR7B
PR7A:   LCALL PR2                      ; SED1520 (2)
PR7B:   MOV   A, R3                    ; 页设置
        INC   A                        ; 页修改
        MOV   R2, A
        CJNE  R6, #00H, PR7C
        LCALL PR0                      ; SED1520 (1)
        LJMP  PR7D
PR7C:   LCALL PR2                      ; SED1520 (2)
PR7D:   MOV   A, R0
        MOV   DPL, A
        MOV   A, R1
        MOV   DPH, A
        MOV   A, R1
        MOV   DPH, A
        MOV   A, R5
        MOVC  A, @A+DPTR
        MOV   R2, A
        CJNE  R6, #00H, PR7E
        LCALL PR1                      ; SED1520 (1)
        LJMP  PR7F
PR7E:   LCALL PR3                      ; SED1520 (2)
PR7F:   INC   R5
        CJNE  R5, #20H, PR7D          ; 下半部数组 (16 字节)
        RET

```

中间结合部汉字位汉字显示处理

```

PR7G:   MOV   R2, #30H                ; SED1520 (1) 列地址
        LCALL PR0                      ; SED1520 (1)
        MOV   R2, #00H                ; SED1520 (2) 列地址
        LCALL PR2                      ; SED1520 (2)
        MOV   A, R3                    ; 页设置
        MOV   R2, A
        LCALL PR0                      ; SED1520 (1)
        LCALL PR2                      ; SED1520 (2)
        MOV   R5, #00H                ; 间址寄存器
        MOV   DPTR, #TAB2              ; 汉字库存首地址
        MOV   A, 61H                   ; 取汉字代码
        MOV   B, #20H
        MUL  AB
        ADD  A, DPL

```



```

MOV R0, A
MOV A, B
ADDC A, DPH
MOV R1, A
PR7H: MOV A, R0
MOV DPL, A
MOV A, R1
MOV DPH, A
MOV A, R5
MOVC A, @A+DPTR ; 取字模
MOV R2, A
CJNE R5, #DB2, PR7I ; MGLS10032: DB2=02H ; MGLS12032: DB2=0CH
PR7I: JNC PR7J
LCALL PR1
LJMP PR7K ; SED1520 (1)
PR7J: LCALL PR3 ; SED1520 (2)
PR7K: INC R5
CJNE R5, #10H, PR7L ; 上半部数组 (16 字节)
MOV R2, #30H ; SED1520 (1) 列地址
LCALL PR0 ; SED1520 (1)
MOV R2, #00H ; SED1520 (2) 列地址
LCALL PR2 ; SED1520 (2)
MOV A, R3 ; 页设置
INC A ; 页地址修改
MOV R2, A
LCALL PR0 ; SED1520 (1)
LCALL PR2 ; SED1520 (2)
PR7L: MOV A, R0
MOV DPL, A
MOV A, R1
MOV DPH, A
MVOV A, R5
MOVC A, @A+DPTR ; 取数据
MOV R2, A
CJNE R5, #DB3, PR7M ; DB3=DB2+10H
PR7M: JNC PR7N
LCALL PR1 ; SED1520 (1)
LJMP PR7O
PR7N: LCALL PR3 ; SED1520 (2)
PR7O: INC R5
CJNE R5, #20H, PR7L ; 下半部数组 (16 字节)
RET

TAB2: DB... ; 汉字库

```